# $Id: PKGBUILD 88699 2013-04-21 22:24:06Z heftig $
# Maintainer:  Ionut Biru <ibiru@archlinux.org>
# Contributor: uastasi <uastasi@archlinux.us>
pkgname=gnote
pkgver=3.8.0
pkgrel=1
pkgdesc="A note taking application."
arch=('i686' 'x86_64')
url="http://live.gnome.org/Gnote"
license=('GPL3')
depends=('gtkmm3' 'libxslt' 'hicolor-icon-theme' 'desktop-file-utils' 'dconf' 'libsecret')
makedepends=('intltool' 'itstool' 'boost')
options=('!libtool')
install=gnote.install
source=(http://ftp.gnome.org/pub/gnome/sources/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.xz)
sha256sums=('603e2cc8782f910a5b4fdfd3588d955d59f0f4ddcd5685d1f2feadb955e0e9e6')

build() {
    cd $pkgname-$pkgver
    ./configure --prefix=/usr \
        --sysconfdir=/etc \
        --localstatedir=/var \
        --libexecdir=/usr/lib/gnote \
        --disable-scrollkeeper \
        --disable-schemas-compile \
        --disable-static
    make
}

package() {
    cd $pkgname-$pkgver
    make DESTDIR="$pkgdir" install
}
